

5.2 Bewegung unter Zentralkräften II

5.2.1 Die Bahn des Planeten Merkur

Merkur ist der sonnennächste Planet und hat nur ein Viertel der Gravitation der Erde. Sein Durchmesser beträgt 4880 km, und er hat einen mittleren Abstand von 58 Millionen km von der Sonne. Die höchste Temperatur beträgt +430 °C, die tiefste etwa -180°C. Auf dem Merkur dauert ein Jahr 88 Erdentage.

Am 3. August 2004 hat die NASA die Sonde *Messenger* auf einen Erkundungsflug in Richtung Merkur geschickt.

"Auf ihrem Weg in den inneren Teil des Sonnensystems gab sie in mehreren [Vorbeiflugmanövern](#) an Erde, Venus und Merkur so viel Bewegungsenergie ab, dass sie am 18. März 2011 beim vierten Vorbeiflug am Merkur mit einem 15-minütigen Bremsmanöver in eine [Umlaufbahn](#) um den Planeten einschwenkte. <http://de.wikipedia.org/wiki/MESSENGER> "

Unser Interesse an Merkur ist die Untersuchung und Aufzeichnung seiner Bahn. In allgemeiner Form haben wir bereits im letzten Kapitel die Grundgleichungen der Bewegungen in einem Gravitationsfeld aufgeführt. In diesem Kapitel werden wir die Differenzialgleichungen des Bewegungsproblems numerisch lösen. Bereits in Kap. 2.5 hatten wir das dazu nötige Programm besprochen und in 2.5.1 entwickelt.

Zunächst haben wir auch hier die wirklichen Variablen in solche ohne Dimension umzuformen. Mit diesen "reduzierten" Variablen können wir dann kleinere Zahlenwerte benutzen, die die numerischen Rechnungen vereinfachen.

Die einzige Kraft, die auf den Planeten einwirkt, ist die Gravitationskraft. Die kartesischen Komponenten dieser Kraft lauten $F_x = m\ddot{x} = -Cm x/r^3$ und $F_y = m\ddot{y} = -Cm y/r^3$.

Für die Beschleunigung erhalten wir die folgenden Gleichungen:

$$x'' = - C x / r^3; \quad y'' = -C y / r^3 \quad (1)$$

hier bedeuten $x'' = d^2x/dt^2$, $C := GM$ und $r=(x^2 + y^2)^{1/2}$

Das System (1) besteht aus zwei gekoppelten Differenzialgleichungen. Vgl. die Kommentare dazu in 2.5.1. (Wir berücksichtigen hier keine Wechselwirkungen (WW) mit anderen Planeten!)

Wie vorhin gesagt, wollen wir einheitenfreie Variablen benutzen. Dies erreichen wir durch Einführung neuer Einheiten: x_0 für Längen und t_0 für Zeiten. Ihre genauen Werte müssen wir noch bestimmen.

Wir schreiben also $x = X \cdot x_0$, $r = R \cdot x_0$ und $t = T \cdot t_0$.

Die neuen Variablen X, R und T haben keine Einheiten.

Die Geschwindigkeit $v = dx/dt$ wird zu $v = dx/dt = x_0/t_0 \cdot dX/dT$, die Beschleunigung ist jetzt $a = d^2x/dt^2 = x_0/t_0^2 \cdot d^2X/dT^2$. $V = dX/dT = v t_0/x_0$.

Die neue Gestalt der Gleichung $x'' = - C x / r^3$ ist

$$d^2x/dt^2 = x_0/t_0^2 \cdot d^2X/dT^2 = - C/x_0^2 \cdot X/R^3 \quad \text{d.h.}$$

$$d^2X/dT^2 = - C t_0^2/x_0^3 \cdot X/R^3 \quad (2)$$

Wir brauchen nur $C t_0^2/x_0^3 := 1$ zu setzen, (3)

um die Bewegungsgleichung ohne Konstanten zu erhalten:

$$d^2X/dT^2 = - X/R^3 \quad (4)$$

Da wir die Bahn des Merkur zeichnen wollen, ist es sinnvoll, x_0 gleich dem Radius der Erdbahn zu wählen, was man *astronomische Einheit AE* nennt:

$$x_0 = 1,496 \cdot 10^{11} \text{m} \quad (5)$$

Unsere neue Zeiteinheit ist

$$t_0 = (x_0^3/C)^{1/2} = 5,027 \cdot 10^6 \text{s} \quad (6)$$

Die Periode des Merkur (Dauer eines Jahres) ist 88 Erdtage. Die Perihel-Daten sind $v_0 = 58,9 \text{ km/s}$ und $r_0 = 46,0 \cdot 10^6 \text{ km}$. Wir nehmen an, dass sich der Planet zur Zeit $T = 0$ im Perihel befindet. Ein Zeitintervall von $\Delta T = 0.05$ bedeutet eine reale Zeit von $\Delta t = \Delta T \cdot t_0 = 0.05 \cdot t_0 = 0.05 \cdot 5,027 \cdot 10^6 \text{s} = 2.91$ Tagen. In der Animation setzen wir alle $dt = \Delta T = 0.005 = 0.291$ Tage ≈ 7 Stunden einen Punkt in den Orbit.

Die *Anfangsbedingungen* lauten:

$$X(0) = x(0)/x_0 = 46 \cdot 10^9 \text{ m}/x_0 = 0,3075$$

$$Y(0) = 0$$

$$dX(0)/dT = 0$$

$$dY(0)/dT = v_{0t_0}/x_0 = 58,9 \cdot 10^3 \text{ m/s} \cdot t_0/x_0 = 1,982$$

Hier folgt nun das animierbare Programm. Es handelt sich um eine Vereinfachung unseres Programms aus 2.5.1.

- `reset() //Merkur-Bahn`

```
DIGITS:=5:
```

```
x0:=0.3075:y0:=0://Anfangsposition
```

```
vx0:=0:// x-Koordinate von v0
```

```
vy0:=1.982:// y-Koordinate von v0
```

```
r3(t):=(x(t)^2+y(t)^2)^(3/2):
```

```
//Differenzialgl'n. mit Anfangsbedingungen
```

```
IVP:={x'(t)=-x(t)/r3(t),y'(t)=-y(t)/r3(t),
```

```
x(0)=x0,x'(0)=vx0,
```

```
y(0)=y0,y'(0)=vy0}:
```

```
fields:=[x(t),y(t),x'(t),y'(t)]:
```

```
ivp:=numeric::ode2vectorfield(IVP, fields):
```

```
Y := numeric::odesolve2(ivp):
```

```
//Tabelle der "reduzierten" Werte:
```

```
print(Unquoted,"T","X","Y");
```

```
for i from 0 to 2 step 0.2 do
```

```
print(i,Y(i)[1],Y(i)[2]):
```

```
end_for;
```

//Animation (Um die Graphik zu sehen, hat man das K.-System zweimal links anzuklicken und dann unter "Animation" *End* zu wählen. Man kann auch den "Start"-Pfeil in der Grafikzeile anklicken.)

```

dt:=0.005:imax1:=280:

plot(

plot::Point2d(Y(t)[1], Y(t)[2],

Color = RGB::Blue,

VisibleFromTo = t..t + 0.99*dt,

PointSize = 2*unit::mm

) $ t in [i*dt $ i = 0..imax1],

plot::Line2d([Y(t - dt)[1], Y(t - dt)[2]],

[Y(t)[1], Y(t)[2]], Color = RGB::Red,

VisibleAfter = t)

$ t in [i*dt $ i = 1..imax1])

```




Fig.: 5.2-1

5.2.2 Streuung von Alphateilchen

Mit dem eben benutzten Programm können wir auch die Bahn bei einer abstoßenden WW aufzeichnen, z.B. die Bahn eines Alphateilchens, das von dem Kern eines Goldatoms abgelenkt wird. In 5.3.4 werden wir genauer auf die Theorie und auf das dazugehörige Experiment eingehen. Hier wollen wir nur eine typische Abstoßungshyperbel aufzeichnen.

In diesem Beispiel haben die Konstanten die folgenden Werte:

$C = 5,6 \text{ m}^3/\text{s}^2$, $r_0 = 14 \cdot 10^{-15} \text{ m} = 14 \text{ F}$ ($1\text{F} = 1 \text{ Fermi} = 1 \cdot 10^{-15} \text{ m}$ wird in der Kernphysik als Längeneinheit benutzt), $v_0 = 2 \cdot 10^7 \text{ m/s}$ (Geschwindigkeit des Teilchens), $t_0 = 7 \cdot 10^{-22} \text{ s}$.

Der Wert von y_0 ist der sogenannte *Stoßparameter* b . In der Figur benutzen wir $y_0 = b = 0,4 \text{ F}$.

- ```

reset()://Alphastreuung
DIGITS:=5:b:=0.4://Stoßparameter
x0:=-5://Ausgangsposition
vx0:=1:// x-Koordinate von v0
vy0:=0:
r3(t):=(x(t)^2+y(t)^2)^(3/2):
// Differenzialgln. mit Anfangsbedingungen
IVP:={x'(t)=x(t)/r3(t),y'(t)=y(t)/r3(t),
x(0)=x0,x'(0)=vx0,
y(0)=b,y'(0)=vy0}:
fields:=[x(t),y(t),x'(t),y'(t)]:
ivp:=numeric::ode2vectorfield(IVP, fields):
Y := numeric::odesolve2(ivp):

//Animation
dt:=0.05:imax:=200:
plot(plot::Point2d(Y(t)[1], Y(t)[2],
Color = RGB::Blue,VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm)
$ t in [i*dt $ i = 0..imax],
plot::Point2d(0,0,Color = RGB::Green,
PointSize = 3*unit::mm),
plot::Line2d([Y(t - dt)[1], Y(t - dt)[2]],
[Y(t)[1], Y(t)[2]], Color = RGB::Red,
VisibleAfter = t) $ t in [i*dt $ i = 1..imax],
ViewingBox =[-5..1,0..5],Scaling = Constrained)

```





Fig.: 5.2-2

Um die Graphik zu sehen, hat man das Koordinaten-System zweimal links anzuklicken und dann unter "Animation" *End* zu wählen. Man kann auch den "Start"-Pfeil in der Grafikzeile anklicken.



### 5.2.3 Bewegung in einem $r^{-1}$ - Feld

Bis jetzt haben wir nur Zentralfelder betrachtet, die umgekehrt proportional zum Quadrat des Abstandes waren –und die Natur wachte streng über die Erhaltung des Exponenten 2. Die Abweichungen von 2 sind geringer als  $2 \cdot 10^{-16}$ . Aber in den Laboratorien können wir Fälle realisieren, bei denen der Exponent sehr verschieden ist von 2, z.B. den Exponenten 1 in einem elektrostatischen Geschwindigkeitsfilter für geladene Teilchen wie Elektronen.

In diesem Abschnitt benutzen wir wieder das Merkur-Programm, allerdings dieses Mal für eine Kraft der Form  $F(r) = k/r$

Wenn man ein Elektron senkrecht in das elektrische Feld einschießt, das sich um einen mit  $q$  Coulomb pro m geladenen geraden, sehr langen Draht bildet, so wird es um den Draht herumfliegen, z.B. auf einer kreisförmigen Bahn, vgl. Fig. 5.2.3 oder 5.2.4.

Aus Symmetriegründen sind die Feldlinien radial und, falls  $q > 0$ , nach außen gerichtet, vgl. 4.5.3. Mithilfe des Gesetzes von Gauss kann man leicht zeigen, dass

$$E(r) = q/(2\pi\epsilon_0) \cdot r^{-1} \quad (7)$$

Die elektrische Feldkonstante  $\epsilon_0$  heißt auch elektrische Permittivität des Vakuums. In SI-Einheiten ist ihr Wert  $8,854 \cdot 10^{-12} \text{ C}^2 \text{ N}^{-1} \text{ m}^{-2}$ .

Für die Kraft, die auf das Elektron einwirkt, gilt

$$F(r) = - qe/(2\pi\epsilon_0) \cdot r^{-1} \quad (8)$$

Die beiden kartesischen Komponenten der Beschleunigung lauten

$$x'' = - C x r^{-2} \quad \text{und} \quad y'' = - C y r^{-2} \quad , \quad (9)$$

worin  $C = q e / (2\pi\epsilon_0 m_e)$  und  $r = (x^2 + y^2)^{1/2}$  bedeuten.

(Vgl. Gln. (9) mit  $x'' = - C x / r^3$ ;  $y'' = -C y / r^3$  (1) !)

$e$  = Elementarladung:  $1,602177 \cdot 10^{-19} \text{ C}$   
 $m_e$  = Elektronenmasse:  $9,10939 \cdot 10^{-31} \text{ kg}$   
 $e/m_e = 1,7588 \cdot 10^{11} \text{ C/kg}$

Damit das Elektron eine Kreisbahn beschreibt, muss gelten  $F(r) = -mv^2/r$ .  
Daraus folgt für die Geschwindigkeit  $v$

$$v = \sqrt{\frac{qe}{2\pi\epsilon_0 m_e}} = \sqrt{C} \quad (10)$$

Daraus ergibt sich, dass Elektronen, die sich auf einer kreisförmigen Bahn bewegen, unabhängig vom Radius dieselbe Geschwindigkeit haben.

Im **ersten Programm** benutzen wir reduzierte Variablen, im **zweiten** nehmen wir Variablen mit normalen Einheiten, also  $s$ ,  $m$  und  $m/s$ .

Im ersten Programm wählen wir als neue Einheiten  $x_0 = 10^{-2} m$  und  $t_0 = 10^{-6} s$ . Eine natürliche Einheit für die Geschwindigkeit wäre  $v_0 = 10^4 m/s$ . Diese Wahl hat zur Folge, dass  $C = 10^8 m^2 s^{-2}$ , siehe Eq. (10). Damit  $C$  diesen Wert annimmt, müssen wir  $q = 3,163 \cdot 10^{-14} C/m$  setzen. Mit diesen Werten erhalten wir dann  $Ct_0^2/x_0^2 = 1$ .

Im ersten Programm bedeutet  $y_0 = 5$  eine Länge von  $5 \cdot 10^{-2} m = 5 cm$ . Die Geschwindigkeit  $vx_0 = 1$  bedeutet eine reale Geschwindigkeit von  $10^4 m/s$ . Das Intervall  $dt = 0.05$  entspricht in Wirklichkeit  $5 \cdot 10^{-8} s$ .

In **Programm 2** benutzen wir  $y_0 = 0.05 m$ ,  $vx_0 = 1E4 m/s$  und  $dt = 5E-8 s$ .

### Programm 1:

```

• reset()://Geladener Draht; reduzierte Variablen
DIGITS:=10:
x0:=0:
y0:=5://5cm
vx0:=1://10000m/s
vy0:=0:
r2(t):=x(t)^2+y(t)^2:
//Differenzialgleichungen
IVP:={x'(t)=-x(t)/r2(t),y'(t)=-y(t)/r2(t),
x(0)=x0,x'(0)=vx0,
y(0)=y0,y'(0)=vy0}:
fields:=[x(t),y(t),x'(t),y'(t)]:
ivp:=numeric::ode2vectorfield(IVP, fields):
Y := numeric::odesolve2(ivp):

```



```

//Animation
dt:=0.05://5·10-8 s
imax:=1000:
plot(
plot::Point2d(Y(t)[1], Y(t)[2], Color =
RGB::Blue,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm) $ t in [i*dt $ i =
0..imax],
plot::Point2d(0,0, Color = RGB::Green,
PointSize = 3*unit::mm),plot::Line2d([Y(t -
dt)[1],
Y(t - dt)[2]], [Y(t)[1], Y(t)[2]], Color =
RGB::Red,
VisibleAfter = t) $ t in [i*dt $ i = 1..imax],
Scaling=Constrained)

```




Fig.: 5.2-3

Wir sehen die Kreisbahn für  $v = 10^4$  m/s. Sie können andere Radien, d.h. andere Werte für  $y_0$  verwenden, z.B.  $y_0 = 1$ , und werden doch jedesmal einen Kreis erhalten.

Wenn Sie aber die Geschwindigkeit ändern, z.B.  $v_{x0} = 1.2$  (= 12 000 m/s), so wird sich eine offene rosettenartige Bahn ergeben:




Fig.: 5.2-4

Das Elektron in Fig. 5.2-4 hatte eine Geschwindigkeit von 12 000 m/s (reduziert = 1.2).

### Programm 2:

In der folgenden Abänderung des vorigen Programms benutzen wir die Variablen in "normaler" Form, d.h. wir geben Längen in m, Zeiten in s und Geschwindigkeiten in m/s ein.

```

• reset();//Geladener Draht; normale Variablen
 DIGITS:=10:
 Q:=3.1648E-14://C/m
 em:=1.7589E11://e/m
 E0:=8.85416E-12:
 C:=Q*em/(2*PI*E0):
 x0:=0:
 y0:=0.05://m
 vx0:=10000://m/s
 vy0:=0:
 r2(t):=x(t)^2+y(t)^2:
 //Differenzialgleichungen
 IVP:={x'(t)=-C*x(t)/r2(t),y'(t)=-C*y(t)/r2(t),
 x(0)=x0,x'(0)=vx0,y(0)=y0,y'(0)=vy0}:
 fields:=[x(t),y(t),x'(t),y'(t)]:
 ivp:=numeric::ode2vectorfield(IVP, fields):
 Y := numeric::odesolve2(ivp):

```

```

//Animation
dt:=5E-8:// s
imax:=1000:
plot(
plot::Point2d(Y(t)[1], Y(t)[2], Color = RGB::Blue,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm) $ t in [i*dt $ i = 0..imax],
plot::Point2d(0,0, Color = RGB::Green,
PointSize = 3*unit::mm),plot::Line2d([Y(t - dt)[1],
Y(t - dt)[2]], [Y(t)[1], Y(t)[2]], Color = RGB::Red,
VisibleAfter = t) $ t in [i*dt $ i = 1..imax],
Scaling=Constrained)

```

Der Graph ist wieder ein Kreis wie in Fig. 5.2-3, nur haben wir auf den Achsen m anstelle von cm, z.B. 0.05 statt 5, usw.

(Benutzen Sie auch andere Anfangswerte, z.B.  $v = 10000$  m/s,  $v_{x0} = 5000$  m/s,  $v_{y0} = 8660$  m/s,  $x_0 = 0.025$  m,  $y_0 = 0.0433$  m. Der Radius ist 0.05 m. Auch mit diesen Daten erhält man einen Kreis.)

Eine besonders ansprechende Figur erhält man mit  $v_{x0} = 10000$  m/s,  $v_{y0} = 8000$  m/s und  $dt = 5E-7$  s :




Fig.: 5.2-5

## Programm zu Fig. 5.2-5

```
reset()// Elektron und geladener Draht.
DIGITS:=10:
Q:=3.1648E-14://C/m
em:=1.7589E11://e/m
E0:=8.85416E-12:
C:=Q*em/(2*PI*E0):
x0:=0:
y0:=0.05://m
vx0:=10000://m/s
vy0:=8000:
r2(t):=x(t)^2+y(t)^2:
//Differenzialgleichungen
IVP:={x'(t)=-C*x(t)/r2(t),y'(t)=-C*y(t)/r2(t),
x(0)=x0,x'(0)=vx0,y(0)=y0,y'(0)=vy0}:
fields:=[x(t),y(t),x'(t),y'(t)]:
ivp:=numeric::ode2vectorfield(IVP, fields):
Y := numeric::odesolve2(ivp):

//"Animation-Ende" benutzen; vgl.S. 5.2-6
dt:=5E-7:// s
imax:=1000:
plot(
plot::Point2d(Y(t)[1], Y(t)[2], Color = RGB::Blue,
VisibleFromTo = t..t + 0.99*dt,
PointSize = 2*unit::mm) $ t in [i*dt $ i = 0..imax],
plot::Point2d(0,0, Color = RGB::Green,
PointSize = 3*unit::mm),plot::Line2d([Y(t - dt)[1],
Y(t - dt)[2]], [Y(t)[1], Y(t)[2]], Color = RGB::Red,
VisibleAfter = t) $ t in [i*dt $ i = 1..imax],
Scaling=Constrained)
```